Ejercicio 2) Disc Jockey

Un Disc Jockey requiere de un proceso que

- a) le permita ingresar una lista de canciones con Nombre y duración en MM:SS, informando al final el tiempo total en HH:MM:SS y La canción con mayor y menor duración
- b) Modifique el ejercicio anterior para que corte en forma automática la carga en el caso que supere los 74 Minutos

Solución:

a)

Análisis:

Entrada:

• Lista de Canciones con nombre y duración de cada canción (MM:SS)

Salida:

- Duración total de la lista de canciones
- Canción de mayor duración
- Canción de menor duración

Procesos:

L = duración de la canción en segundos

- Canción mayor duración L = { L >= d, ∀ d / d y L ∈ duración de todas las canciones ingresadas}
- Canción menor duración L = { L <= d, ∀ d / d y L ∈ duración de todas las canciones ingresadas}

Estrategia:

- 1. Solicitamos ingresar el nombre de la primer canción
- 2. Solicitamos ingresar la duración de la primer canción
- 3. Asignamos el nombre y duración de la primer canción como la más corta
- 4. Asignamos el nombre y duración de la primer canción como la más larga
- 5. Asignamos la duración de la canción como la duración total de las canciones
- 6. Preguntamos si se desean ingresar más canciones
- 7. Mientras se deseen ingresar más canciones:
 - a. Solicitamos ingresar el nombre de la siguiente canción
 - b. Solicitamos ingresar la duración de la siguiente canción
 - c. Sumamos la duración de la canción a la duración total de las canciones
 - d. Verificamos si la duración de la canción es menor a la duración de la asignada como *la más corta*.
 - i. Si se verifica, asignamos el nombre y duración de la canción como *La más corta.*
 - e. Verificamos si la duración de la canción es mayor a la duración de la asignada como *la más larga*.
 - i. Si se verifica, asignamos el nombre y duración de la canción como La más larga.
 - f. Preguntamos si se desean ingresar más canciones
- 8. Mostramos la canción guardada como La más corta.
- 9. Mostramos la canción guardada como La más larga.
- 10. Mostramos la duración total de las canciones.

Ambiente:

Variable	Tipo	Descripción
dTotal	Entero	Duración total en segundos de todas las canciones
dLarga	Entero	Duración en segundos de la canción más larga
dCorta	Entero	Duración en segundos de la canción más corta
dHora	Entero	Cantidad de horas de la duración total
dMin	Entero	Cantidad de minutos de la duración total
dSeg	Entero	Duración total de segundos de la canción
min	Entero	Cantidad de Minutos de la canción
seg	Entero	Cantidad de seg de la canción
nombre	Caracter	Nombre de la canción
cad	Caracter	Duración de la canción en formato HH:SS
nLarga	Caracter	Nombre de la canción más larga
nCorta	Caracter	Nombre de la canción más corta
орс	Caracter	Opción seleccionada

Algoritmo:

```
Algoritmo ejercicio2
      Definir min, seq, dSeq, dMin, dHora, dCorta, dLarga, dTotal
Como Entero
      Definir nombre, nLarga, nCorta, cad, opc Como Caracter
      //Iniciamos la duración total de las canciones en 0
      dTotal <- 0
      //Solicitamos la primer canción
      Escribir 'Ingrese el Nombre de la canción'
      Leer nombre
      Escribir 'Ingrese la duración de la canción (MM:SS)'
      Leer cad
      //Convertimos el formato texto a la duración total de la
canción en segundos
      min <- ConvertirANumero(Subcadena(cad,0,2))</pre>
      seg <- ConvertirANumero(Subcadena(cad,4,5))</pre>
      dSeg <- min*60 + seg
      // Agregamos la duración de la canción a la canción total
      dTotal <- dTotal + dSeg
      //Asignamos la primer canción como la más larga y la más corta
      nLarga <- nombre + " " + cad
      dLarga <- dSeg
      nCorta <- nombre + " " + cad
      dCorta <- dSeg
      // Continuamos ingresando canciones?
      Escribir "¿Desea ingresar otra canción? (S/N)"
      Leer opc;
      Mientras ( opc == "S" | opc == "s")
            Escribir 'Ingrese el Nombre de la canción'
            Leer nombre
            Escribir 'Ingrese la duración de la canción (MM:SS)'
            Leer cad
            //Convertimos el formato texto a la duración total de la
canción en segundos
            min <- ConvertirANumero(Subcadena(cad,0,2))</pre>
            seg <- ConvertirANumero(Subcadena(cad,4,5))</pre>
            dSeg <- min*60 + seg
            // Agregamos la duración de la canción a la canción total
            dTotal <- dTotal + dSeg
            //Comprobamos si es más larga que la guardada como más
larga
            Si (dSeg > dLarga) Entonces
                  nLarga <- nombre + " " + cad
                  dLarga <- dSeg
            FinSi
            //Comprobamos si es más corta que la guardada como más
corta
            Si (dSeg < dCorta) Entonces
                  nCorta <- nombre + " " + cad
```

```
dCorta <- dSeg
           FinSi
           Escribir 'Canción ', nombre, ' - ', min, ':', seg
           Escribir "¿Desea ingresar otra canción? (S/N)"
           Leer opc;
     FinMientras
     //Mostramos la canción más corta
     Escribir 'Canción más corta: ', nCorta
     //Mostramos la canción más larga
     Escribir 'Canción más Larga: ', nLarga
     //Formateamos la duracion en segundos
     dHora <- trunc(dTotal/3600) // horas en el total de segundos
     dTotal <- dTotal mod 3600 //resto
     dMin <- trunc(dTotal/60) // minutos en el total de segundos</pre>
     dSeg <- dTotal mod 60 //resto
     //Mostramos la duración total de las canciones
     Escribir
               "Duración Total
                                       de
                                            las canciones: ",
dHora, ":", dMin, ":", dSeg;
FinAlgoritmo
```